// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.19;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.19;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.19;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: src/AlohaPresale.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

import "@oz/access/Ownable.sol";
import "@oz/token/ERC20/IERC20.sol";

contract AlohaPresale is Ownable {
    /// @notice The token being sold
    IERC20 public token;

    /// @notice Total supply of tokens for sale
    uint256 public supply;

    /// @notice Minimum amount a buyer can purchase
    uint256 public minAmount;

    /// @notice Maximum amount a buyer can purchase
    uint256 public maxAmount;

    /// @notice Total ETH collected from the sale
    uint256 public totalSales;

    /// @notice A structure to keep track of each buyer
    struct Buyer {
        address addr; // Address of the buyer
        uint256 amount; // Amount of ETH contributed by the buyer
    }

    /// @notice Mapping from index to Buyer struct
    mapping(uint256 => Buyer) public buyer;

    /// @notice Mapping from address of buyer to the index in the 'buyer' mapping
    mapping(address => uint256) public buyerIndex;

    /// @notice Number of unique buyers in the sale
    uint256 public totalBuyers;

    /// @notice Initializes contract with the creator as the owner
    constructor() Ownable(msg.sender) {}

    /// @notice Buy into the presale
    function buy() public payable {
        _presaleBuy();
    }

    /// @dev Fallback function for receiving ETH
    receive() external payable {
        _presaleBuy();
    }

    /// @dev Handles token purchase
    function _presaleBuy() private {
        require(maxAmount != 0, "Sale Not Open");
        require(msg.value >= minAmount, "Below ETH Minimum");
        uint256 index = buyerIndex[msg.sender];
        if (index == 0) {
            index = ++totalBuyers;
        }
        uint256 total = buyer[index].amount;
        require(msg.value + total <= maxAmount, "Above ETH Maximum");
        buyerIndex[msg.sender] = index;
        buyer[index].addr = msg.sender;
        buyer[index].amount += msg.value;
        totalSales += msg.value;
    }

    /// @dev Implements the logic for distributing tokens to buyers
    function _airdrop(uint256 from, uint256 to) private {
        require(address(token) != address(0) && supply != 0);
        uint256 tokenPrice = (totalSales * 10e18) / supply;
        for (uint256 i = from; i <= to; i++) {
            uint256 amount = buyer[i].amount * 10e18;
            uint256 numTokens = amount / tokenPrice;
            token.transfer(buyer[i].addr, numTokens);
        }
    }

    /// @dev Implements the logic for refunding buyers
    function _refund(uint256 a, uint256 b) private {
        require(a > 0);
        for (uint256 i = a; i <= b; i++) {
            address payable refunding = payable(buyer[i].addr);
            refunding.transfer(buyer[i].amount);
        }
    }

    /// @notice Allows contract owner to distribute tokens to the buyers
    /// @dev This function should be called after the sale has ended. It will calculate and distribute the tokens based on the contribution of each buyer.
    /// @dev Only accessible by contract owner
    function airdrop() external onlyOwner {
        _airdrop(1, totalBuyers);
    }

    /// @notice Allows contract owner to distribute tokens to a range of buyers
    /// @param a Index to start airdrop from
    /// @param b Index to stop airdrop at
    /// @dev This function should be called after the sale has ended. It will calculate and distribute the tokens based on the contribution of each buyer in the specified range.
    /// @dev Only accessible by contract owner
    function airdrop(uint256 a, uint256 b) external onlyOwner {
        require(a > 0);
        _airdrop(a, b);
    }

    /// @notice Allows contract owner to refund buyers
    /// @dev This function should be called if the sale has to be cancelled. It will refund the contributed Ether back to the buyers.
    /// @dev Only accessible by contract owner
    function refund() external onlyOwner {
        _refund(1, totalBuyers);
    }

    /// @notice Allows contract owner to refund a range of buyers
    /// @param from Index to start refund from
    /// @param to Index to stop refund at
    /// @dev This function should be called if the sale has to be cancelled. It will refund the contributed Ether back to the buyers in the specified range.
    /// @dev Only accessible by contract owner
    function refund(uint256 from, uint256 to) external onlyOwner {
        require(from > 0);
        _refund(from, to);
    }

    /// @notice Allows the contract owner to withdraw collected ETH
    /// @dev Only accessible by contract owner
    function withdraw() external onlyOwner {
        (bool success, ) = payable(owner()).call{value: address(this).balance}(
            ""
        );
        require(success, "Withdraw Failed");
    }

    /// @notice Allows the contract owner to withdraw any ERC20 tokens sent to this contract by mistake
    /// @dev Only accessible by contract owner
    function withdrawERC20() external onlyOwner {
        token.transfer(owner(), token.balanceOf(address(this)));
    }

    /// @notice Allows contract owner to set the token for the sale and its supply
    /// @param token_ The address of the token contract
    /// @param supply_ Total supply of the token for sale
    /// @dev Only accessible by contract owner
    function updateToken(IERC20 token_, uint256 supply_) external onlyOwner {
        token = token_;
        supply = supply_;
    }

    /// @notice Allows contract owner to update minimum and maximum purchase limits
    /// @param min New minimum purchase limit
    /// @param max New maximum purchase limit
    /// @dev Only accessible by contract owner
    function updateMinMaxAmount(uint256 min, uint256 max) public onlyOwner {
        minAmount = min;
        maxAmount = max;
    }
}